//

NUMERICAL STUDIES OF VISCOUS, INCOMPRESSIBLE FLOW FOR ARBITRARY REYNOLDS NUMBER

by Donald Greenspan

APPENDIX: PROGRAMMING VISCOUS, INCOMPRESSIBLE FLOW PROBLEMS

by M. McClellan

Computer Sciences Technical Report #11
January 1968

NUMERICAL STUDIES OF VISCOUS, INCOMPRESSIBLE FLOW
FOR ARBITRARY REYNOLDS NUMBER

by

Donald Greenspan

1. Introduction

The development of the high speed digital computer has resulted in extensive efforts to solve numerically fluid problems whose equations of motion are the Navier-Stokes equations (see, e.g., references [1]-[5] and the additional references contained therein). The interest in these equations is founded not only on the fact that they incorporate boundary layer phenomena, but also on the important observation that they result from both microscopic and macroscopic approaches to viscous flow [6], [7].

In this paper we will adapt a new numerical method [3] to study a prototype problem of the circulation of a steady, viscous, incompressible flow within a square boundary. Both the stream function and its normal derivative will be prescribed on the boundary. The discussion will be self contained and the numerical method will apply equally well to comparable boundary value problems. We will consider with equal ease cases in which the Reynolds number is small ($\Re = 10$) and cases in which the

Reynolds number is large ($\mathfrak{R} = 10^5$). If and when such steady state flows exist, which is still usually an open matter, the method to be described is vastly more economical and accurate than time dependent, step-by-step methods. The power of our method is contained in the structure of the difference equations which, for <u>all</u> $\mathfrak{R}$, yield diagonally dominant systems of linear algebraic equations.

2. Statement of the Analytical Problem.

The problem to be considered can be formulated as follows. Let the points (0, 0), (1, 0), (1, 1) and (0, 1) be denoted by A, B, C and D, respectively (see Figure 2.1). Let S be the square whose vertices are A, B, C, D and denote its interior by R. On R the equations of motion to be satisfied are the two dimensional, steady state, Navier-Stokes equations, that is

$$\Delta\psi = -\omega \tag{2.1}$$

$$\Delta\omega + \mathfrak{R}\left(\frac{\partial\psi}{\partial x}\frac{\partial\omega}{\partial y} - \frac{\partial\psi}{\partial y}\frac{\partial\omega}{\partial x}\right) = 0\ , \tag{2.2}$$

where ψ is the stream function, ω is the vorticity, and $\mathfrak{R}$ is the Reynolds number. On S the boundary conditions tobe satisfied are

$$\psi = 0, \qquad \frac{\partial\psi}{\partial x} = 0\ , \quad \text{on AD} \tag{2.3}$$

$$\psi = 0, \qquad \frac{\partial\psi}{\partial y} = 0\ , \quad \text{on AB} \tag{2.4}$$

$$\psi = 0, \qquad \frac{\partial\psi}{\partial x} = 0\ , \quad \text{on BC} \tag{2.5}$$

$$\psi = 0, \qquad \frac{\partial\psi}{\partial y} = -1\ , \quad \text{on CD}\ . \tag{2.6}$$

The analytical problem is defined on R + S by (2.1) - (2.6) and is shown diagramatically in Figure 2.1.



Figure 2.1

3. Difference Approximations.

Because the numerical method to be developed will be a finite difference method, it will be convenient in this section to recall or to develop several useful finite difference approximations.

First, for h $>$ 0, consider the five points (x, y), (x+h, y), (x, y+h), (x-h, y) and (x, y-h), numbered 0, 1, 2, 3 and 4, respectively, in Figure 3.1. For convenience, any function u(x, y) defined at a point numbered i will be denoted at that point by u_i . Now, if $\omega(x, y)$ is defined at the point numbered 0 in Figure 3.1, then (2.1) can be approximated at 0 by the well known [8] Poisson difference analogue

(3.1) $$-4\,\psi_0 + \psi_1 + \psi_2 + \psi_3 + \psi_4 = -h^2\,\omega_0 \,.$$



Figure 3.1

On the other hand, if $\psi(x, y)$ is defined at the points numbered 0, 1, 2, 3, 4 in Figure 3.1, then (2.2) can be approximated [3] as follows. Set

$$\alpha = \psi_1 - \psi_3$$

$$\beta = \psi_2 - \psi_4$$

and at the point numbered 0 in Figure 3.1 approximate (2.2) by

(3.2a) $$\left(-4 - \frac{\alpha\mathfrak{R}}{2} - \frac{\beta\mathfrak{R}}{2}\right)\omega_0 + \omega_1 + \left(1 + \frac{\alpha\mathfrak{R}}{2}\right)\omega_2 + \left(1 + \frac{\beta\mathfrak{R}}{2}\right)\omega_3 + \omega_4 = 0,$$

if $\alpha \geq 0$, $\beta \geq 0$;

(3.2b) $$\left(-4 - \frac{\alpha\mathfrak{R}}{2} + \frac{\beta\mathfrak{R}}{2}\right)\omega_0 + \left(1 - \frac{\beta\mathfrak{R}}{2}\right)\omega_1 + \left(1 + \frac{\alpha\mathfrak{R}}{2}\right)\omega_2 + \omega_3 + \omega_4 = 0,$$

if $\alpha \geq 0$, $\beta < 0$;

(3.2c) $$\left(-4 + \frac{\alpha\mathfrak{R}}{2} - \frac{\beta\mathfrak{R}}{2}\right)\omega_0 + \omega_1 + \omega_2 + \left(1 + \frac{\beta\mathfrak{R}}{2}\right)\omega_3 + \left(1 - \frac{\alpha\mathfrak{R}}{2}\right)\omega_4 = 0,$$

if $\alpha < 0$, $\beta \geq 0$;

(3.2d) $$\left(-4 + \frac{\alpha\mathfrak{R}}{2} + \frac{\beta\mathfrak{R}}{2}\right)\omega_0 + \left(1 - \frac{\beta\mathfrak{R}}{2}\right)\omega_1 + \omega_2 + \omega_3 + \left(1 - \frac{\alpha\mathfrak{R}}{2}\right)\omega_4 = 0,$$

if $\alpha < 0$, $\beta < 0$.

Next, recall that for three points (x, y), $(x+h, y)$, $(x+2h, y)$, numbered 0, 1, 2, respectively, in Figure 3.2(a), one has the approximation [9]

(3.3a) $$\left.\frac{\partial\psi}{\partial x}\right|_0 = \frac{1}{2h}(-3\psi_0 + 4\psi_1 - \psi_2) \; ;$$

for three points (x, y), $(x, y+h)$, $(x, y+2h)$, numbered 0, 1, 2, respectively, in Figure 3.2(b), one has the approximation

(3.3b) $$\left.\frac{\partial\psi}{\partial y}\right|_0 = \frac{1}{2h}(-3\psi_0 + 4\psi_1 - \psi_2) \; ;$$

for three points (x, y), $(x-h, y)$, $(x-2h, y)$, numbered $0, 1, 2$, respectively, in Figure 3.2(c), one has the approximation

$$\left.\frac{\partial\psi}{\partial x}\right|_0 = \frac{1}{2h}(3\psi_0 - 4\psi_1 + \psi_2) \; ; \tag{3.3c}$$

and that for three points (x, y), $(x, y-h)$, $(x, y-2h)$, numbered $0, 1, 2$, respectively, in Figure 3.2(d), one has

$$\left.\frac{\partial\psi}{\partial y}\right|_0 = \frac{1}{2h}(3\psi_0 - 4\psi_1 + \psi_2) \; . \tag{3.3d}$$

Finally, let us develop approximations for the Laplace operator $\psi_{xx} + \psi_{yy}$ on S in terms of certain function values and normal derivatives.



Figure 3.2



Figure 3.3

Consider the four points (x, y), $(x+h, y)$, $(x, y+h)$, $(x, y-h)$, numbered 0, 1, 2, 4, respectively, in Figure 3.3(a), and let us try to determine parameters $\alpha_0, \alpha_1, \alpha_2, \alpha_4, \alpha_5$ such that

(3.4) $$(\psi_{xx} + \psi_{yy})\big|_0 = \alpha_0 \psi_0 + \alpha_1 \psi_1 + \alpha_2 \psi_2 + \alpha_4 \psi_4 + \alpha_5 \left(\frac{\partial \psi}{\partial x}\right)\bigg|_0 .$$

In (3.4), expansion of ψ_1, ψ_2 and ψ_4 into Taylor series about the point numbered 0 and reorganization of terms implies

$$\begin{aligned}(\psi_{xx} + \psi_{yy})\big|_0 = {} & \psi_0 (\alpha_0 + \alpha_1 + \alpha_2 + \alpha_4) \\ & + \psi_x (h\alpha_1 + \alpha_5) \\ & + \psi_y (h\alpha_2 - h\alpha_4) \\ & + \psi_{xx} \left(\frac{h^2}{2} \alpha_1\right) \\ & + \psi_{yy} \left(\frac{h^2}{2} \alpha_2 + \frac{h^2}{2} \alpha_4\right) \\ & + \cdots\cdots .\end{aligned}$$

In this latter equality, the setting of corresponding terms equal yields

$$\begin{aligned}\alpha_0 + \alpha_1 + \alpha_2 + \alpha_4 &= 0 \\ h\alpha_1 + \alpha_5 &= 0 \\ h\alpha_2 - h\alpha_4 &= 0 \\ \frac{h^2}{2} \alpha_1 &= 1 \\ \frac{h^2}{2} \alpha_2 + \frac{h^2}{2} \alpha_4 &= 1 ,\end{aligned}$$

the solution of which is

$$\alpha_0 = -\frac{4}{h^2}, \quad \alpha_1 = \frac{2}{h^2}, \quad \alpha_2 = \alpha_4 = \frac{1}{h^2}, \quad \alpha_5 = -\frac{2}{h} .$$

Thus one arrives at the following approximation:

$$(3.5a) \qquad (\psi_{xx} + \psi_{yy})\big|_0 = -\frac{4}{h^2}\psi_0 + \frac{2}{h^2}\psi_1 + \frac{1}{h^2}\psi_2 + \frac{1}{h^2}\psi_4 - \frac{2}{h}\left(\frac{\partial\psi}{\partial x}\right)\Big|_0 .$$

Similarly, for the four points (x, y), (x+h, y), (x, y+h), (x-h, y), numbered 0, 1, 2, 3, respectively, in Figure 3.3(b), one has

$$(3.5b) \qquad (\psi_{xx} + \psi_{yy})\big|_0 = -\frac{4}{h^2}\psi_0 + \frac{1}{h^2}\psi_1 + \frac{2}{h^2}\psi_2 + \frac{1}{h^2}\psi_3 - \frac{2}{h}\left(\frac{\partial\psi}{\partial y}\right)\Big|_0 ;$$

for the four points (x, y), (x, y+h), (x-h, y), (x, y-h), numbered 0, 2, 3, 4, respectively, in Figure 3.3(c), one has

$$(3.5c) \qquad (\psi_{xx} + \psi_{yy})\big|_0 = -\frac{4}{h^2}\psi_0 + \frac{1}{h^2}\psi_2 + \frac{2}{h^2}\psi_3 + \frac{1}{h^2}\psi_4 + \frac{2}{h}\left(\frac{\partial\psi}{\partial x}\right)\Big|_0 ;$$

and for the four points (x, y), (x+h, y), (x-h, y), (x, y-h), numbered 0, 1, 3, 4, respectively, in Figure 3.3(d), one has

$$(3.5d) \qquad (\psi_{xx} + \psi_{yy})\big|_0 = -\frac{4}{h^2}\psi_0 + \frac{1}{h^2}\psi_1 + \frac{1}{h^2}\psi_3 + \frac{2}{h^2}\psi_4 + \frac{2}{h}\left(\frac{\partial\psi}{\partial y}\right)\Big|_0 .$$

Note that the numbering of the points in Figure 3.3 is consistent with that in Figure 3.1 .

4. The Numerical Method.

For a fixed positive integer n, set $h = \frac{1}{n}$. Starting at (0, 0) with grid size h, construct and number in the usual way [8] the set of interior grid points R_h and the set of boundary grid points S_h. To within some preassigned tolerance ϵ, we aim to find a solution $\psi^{(k)}$ of (3.1) on R_h and a solution $\omega^{(k)}$ of (3.2a) - (3.2d) on $R_h + S_h$, subject to the boundary restrictions on ψ, and we proceed as follows.

Denote by $R_{h,1}$ those points of R_h whose distance from S is h, and denote by $R_{h,2}$ those points of R_h whose distance from S is greater than h. Initially, set

(4.1) $$\psi^{(0)} = C_1 \quad , \text{ on } R_h$$

(4.2) $$\omega^{(0)} = C_2 \quad , \text{ on } R_h + S_h ,$$

where C_1 and C_2 are constants. A modified over-relaxation procedure which does not require much storage to obtain the desired result is then applied as follows to yield $\psi^{(1)}$ from $\psi^{(0)}$ and $\omega^{(0)}$. On R_h, set

(4.3) $$\psi^{(1,0)} = \psi^{(0)}$$

and on $R_{h,2}$ generate $\overline{\psi}^{(1,1)}$ by sweeping along each row of $R_{h,2}$ from left to right, starting from the bottom row and proceeding to the top row, by the recursion formula

(4.4) $$\overline{\psi}_0^{(1,j)} = (1-r_\psi)\,\psi_0^{(1,j-1)} + \frac{r_\psi}{4}\,[\psi_1^{(1,j-1)} + \psi_2^{(1,j-1)} + \overline{\psi}_3^{(1,j)} + \overline{\psi}_4^{(1,j)}] ,$$

where $0 < r_\psi < 1$. After each such sweep, $\psi^{(1,j)}$ is defined on $R_{h,2}$ by the weighted average

(4.5) $$\psi^{(1,j)} = \xi\,\psi^{(1,j-1)} + (1-\xi)\,\overline{\psi}^{(1,j)} , \quad 0 \le \xi \le 1 .$$

This inner iteration process continues until, for the given tolerance ϵ, one has

$$|\psi^{(1,k)} - \psi^{(1,k+1)}| < \epsilon \quad , \tag{4.6}$$

from which one defines on $R_{h,2}$

$$\psi^{(1)} = \psi^{(1,k)} \quad . \tag{4.7}$$

In order to define $\psi^{(1)}$ on $R_{h,1}$, we apply (3.3a) - (3.3d) and (2.3) - (2.6) in the following fashion. At each point of $R_{h,1}$ of the form (ih, h), $i = 1, 2, \ldots, n-1$, set (in the notation of Figure 3.2(b))

$$\psi_1^{(1)} = \frac{\psi_2^{(1)}}{4} \quad . \tag{4.8a}$$

Similarly, at each point of $R_{h,1}$ of the form (h, ih), $i = 2, 3, \ldots, n-2$, set (in the notation of Figure 3.2(a))

$$\psi_1^{(1)} = \frac{\psi_2^{(1)}}{4} \quad , \tag{4.8b}$$

while at each point of $R_{h,1}$ of the form $(1-h, ih)$, $i = 2, 3, \ldots, n-2$, set (in the notation of Figure 3.2(c))

$$\psi_1^{(1)} = \frac{\psi_2^{(1)}}{4} \quad . \tag{4.8c}$$

Finally, at each point of $R_{h,1}$ of the form $(ih, 1-h)$, $i = 1, 2, \ldots, n-1$, set (in the notation of Figure 3.2(d))

$$\psi_1^{(1)} = \frac{h}{2} + \frac{\psi_2^{(1)}}{4} \quad . \tag{4.8d}$$

Thus, (4.3) and (4.4a) - (4.4d) define $\psi^{(1)}$ on all of R_h .

Next, proceed to construct $\omega^{(1)}$ on $R_h + S_h$ as follows. On S_h , use (2.1), (2.3)-(2.6) and (3.5a) - (3.5d) to yield at each point (ih, 0), $i = 0, 1, 2, \ldots, n$ (in the notation of Figure 3.3(b))

(4.9a) $$\overline{\omega}_0^{(1)} = -\frac{2\psi_2^{(1)}}{h^2}$$

at each point (0, ih), $i = 1, 2, \ldots, n-1$, in the notation of Figure 3.3(a)

(4.9b) $$\overline{\omega}_0^{(1)} = -\frac{2\psi_1^{(1)}}{h^2} ;$$

at each point (1, ih), $i = 1, 2, \ldots, n-1$, in the notation of Figure 3.3(c)

(4.9c) $$\overline{\omega}_0^{(1)} = -\frac{2\psi_3^{(1)}}{h^2}$$

and, at each point (ih, 1), $i = 0, 1, 2, \ldots, n$, in the notation of Figure 3.3(d)

(4.9d) $$\overline{\omega}_0^{(1)} = \frac{2}{h} - \frac{2\psi_4^{(1)}}{h^2} .$$

One then defines $\omega^{(1)}$ on S_h by the weighted average formula

(4.10) $$\omega^{(1)} = \delta\,\omega^{(0)} + (1 - \delta)\,\overline{\omega}^{(1)} , \quad 0 \le \delta \le 1 .$$

We proceed next to determine $\omega^{(1)}$ on R_h by again using a modified over-relaxation procedure. At each point of S_h set

$$\omega^{(1,0)} = \omega^{(1)}$$

while at each point of R_h set

$$\omega^{(1,0)} = \omega^{(0)} .$$

Then generate $\overline{\omega}^{(1,1)}$ by sweeping along each row of R_h from left to right, starting from the bottom row and proceeding to the top row, by the recursion formula

$$(4.11) \qquad \overline{\omega}^{(1,j)} = (1 - r_\omega)\,\omega^{(1,j-1)} + \frac{r_\omega}{\Omega_0}\left[\Omega_1 \cdot \omega_1^{(1,j-1)} + \Omega_2 \cdot \omega_2^{(1,j-1)} + \Omega_3 \cdot \overline{\omega}_3^{(1,j)} + \Omega_4 \overline{\omega}_4^{(1,j)}\right] ,$$

where $0 < r_\omega < 2$, where

$$\Omega_0 = 4 + \frac{\Re}{2}|\alpha| + \frac{\Re}{2}|\beta|$$

$$\Omega_1 = \begin{cases} 1 & , \ \beta \geq 0 \\ 1 + \frac{\Re}{2}|\beta| & , \ \beta < 0 , \end{cases}$$

$$\Omega_2 = \begin{cases} 1 + \frac{\Re}{2}\alpha & , \ \alpha \geq 0 \\ 1 & , \ \alpha < 0 \end{cases}$$

$$\Omega_3 = \begin{cases} 1 + \frac{\Re}{2}\beta & , \ \beta \geq 0 \\ 1 & , \ \beta < 0 \end{cases}$$

$$\Omega_4 = \begin{cases} 1 & , \ \alpha \geq 0 \\ 1 + \frac{\Re}{2}|\alpha| & , \ \alpha < 0 \end{cases}$$

and where, as defined previously,

$$\alpha = \psi_1 - \psi_3$$

$$\beta = \psi_2 - \psi_4 \quad .$$

After each such sweep, $\omega^{(1, j)}$ is defined on R_h by the weighted average

$$\omega^{(1, j)} = \delta\, \omega^{(1, j-1)} + (1 - \delta)\, \overline{\omega}^{(1, j)}, \quad 0 \le \delta \le 1 \tag{4.12}$$

where δ is the same weight as that used in (4.10). This inner iteration continues until, for the given tolerance ϵ, one has

$$|\omega^{(1,K)} - \omega^{(1,K+1)}| < \epsilon \quad , \tag{4.13}$$

from which one defines on R_h

$$\omega^{(1)} = \omega^{(1,K)} \quad . \tag{4.14}$$

Proceed next to determine $\psi^{(2)}$ on R_h from $\omega^{(1)}$ and $\psi^{(1)}$ in the same fashion as $\psi^{(1)}$ was determined from $\omega^{(0)}$ and $\psi^{(0)}$. Then construct $\omega^{(2)}$ on $R_h + S_h$ from $\omega^{(1)}$ and $\psi^{(2)}$ in the same fashion as $\omega^{(1)}$ was determined from $\omega^{(0)}$ and $\psi^{(1)}$. In the indicated fashion, construct the finite sequences of outer iterates

$$\psi^{(0)}, \psi^{(1)}, \psi^{(2)}, \ldots\ldots, \psi^{(m)}$$

$$\omega^{(0)}, \omega^{(1)}, \omega^{(2)}, \ldots\ldots, \omega^{(m)}$$

which satisfy

$$|\psi^{(m)} - \psi^{(m+1)}| < \epsilon \quad , \quad \text{on } R_h$$

$$|\omega^{(m)} - \omega^{(m+1)}| < \epsilon \quad , \quad \text{on } R_h + S_h \, .$$

The discrete functions $\psi^{(m)}$ and $\omega^{(m)}$ are taken to be the numerical approximations of $\psi(x, y)$ and $\omega(x, y)$, respectively, after verifying that they satisfy (3.1) and (3.2a)-(3.2d).

5. Examples.

We will attempt now to summarize the results of the large number of examples run on the CDC 3600 at the University of Wisconsin.

In Figures 5.1 - 5.12 are shown graphically the streamlines and equivorticity curves for $\Re = 10, 100, 500, 1000, 3000, 100000$ for the set of parameter values $h = \frac{1}{20}$, $C_1 = C_2 = 0$, $r_\psi = 1.8$, $r_\omega = 1$, $\xi = 0.1$, $\delta = 0.7$. A tolerance of 10^{-4} was taken for convergence of both inner and outer iterations. The outer iterations for each of $\Re = 10, 500, 1000, 3000, 100000$ converged in fewer than ten minutes and the number of outer interations required were, respectively, 10, 25, 20, 16, 14. The case $\Re = 100$ was allowed only twelve minutes of running time at the end of which 40 iterations had elapsed and convergence to $6 \cdot 10^{-4}$ had resulted.

It was clear that for convergence ξ and δ depended on h. For fixed $C_1 = C_2 = 0$ and $r_\psi = 1.8$, $r_\omega = 1$, the following results were found. Outer iteration convergence was achieved for $h = \frac{1}{8}$, $\xi = \delta = 0$, but outer iteration divergence resulted in every case for $h \le \frac{1}{10}$, $\xi = \delta = 0$. For

$h = \frac{1}{16}$, outer iteration convergence was achieved with the choice $\xi = \delta = 0.1$, but outer iteration convergence was greatly accelerated as δ was allowed to increase. For $h = \frac{1}{20}$, all choices of $\xi \le 0.1$, $\delta \le 0.5$ resulted in outer iteration divergence. Further experimentation into the relationships between ξ, δ, h, convergence, and divergence was deemed to be of great interest but too costly to be run at the present time.

Occasionally, the method did not converge because an inner iteration did not converge. When this happened, invariably the choice of r_ω was at fault and a new choice was made after several trial values were tested. The choices $r_\psi = 1.8$ and $r_\omega = 1$ were finally decided upon because they worked well uniformly, even though inner iteration convergence could often be accelerated by different choices.

In cases where the outer iterations were diverging, no choices of C_1 and C_2 ever resulted in convergence.

With regard to the physics of the problem, it should be observed that Figures 5.7 - 5.12 indicate clearly that the vorticity is becoming uniform in a large connected subregion of the given region, as was predicted theoretically by Batchelor [10].

Finally, it should be noted that we are documenting our computations by the inclusion of the computer program in an appendix. This is absolutely necessary if other workers in the field are to be able to duplicate our computations in order to verify or to refute our results. Such an omission in the paper of Burggraf [1] caused us great consternation since our duplication of his work for $R = 0$ yielded divergence while he claimed convergence. In this connection, the recent report of Smith [11] proves theoretically that Burggraf's method must diverge for all sufficiently small h .



FIGURE 5.1 Streamlines for Reynolds number 10.



FIGURE 5.2 Streamlines for Reynolds number 100



FIGURE 5.3 Streamlines for Reynolds number 500



FIGURE 5.4 Streamlines for Reynolds number 1000



FIGURE 5.5 Streamlines for Reynolds number 3000



FIGURE 5.6 Streamlines for Reynolds number 100000



FIGURE 5.7 Equivorticity curves for Reynolds number 10.



FIGURE 5.8 Equivorticity curves for Reynolds number 100



FIGURE 5.9 Equivorticity curves for Reynolds number 500



FIGURE 5.10 Equivorticity curves for Reynolds number 1000



FIGURE 5.11 Equivorticity curves for Reynolds number 3000



FIGURE 5.12 Equivorticity curves for Reynolds number 100000

APPENDIX

PROGRAMMING VISCOUS, INCOMPRESSIBLE FLOW PROBLEMS

by M. McClellan

Definitions of Main Program Variables and Parameters

PSI = stream function vector

W = vorticity vector

XMAX, XMIN = extreme values of x for rectangular region

YMAX, YMIN = " " " y " " "

R = Reynold's number

H = grid size

OMEGAP = relaxation factor for PSI inner-iterations

OMEGAW = " " " W " "

TOL = tolerance for both inner- and outer-iterations

XI = weighting factor for PSI

DELTA = " " " W

M = number of vertical lines in the grid

N = " " horizontal " " " "

ITERMAX = maximum number of outer-iterations

ITERMAXP = " " " inner-iterations for PSI.

ITERMAXW = " " " " " " W

TOLTEST = number of outer-iterations between tests for problem convergence

TOLTESTP = number of PSI inner-iterations between tests for convergence

TOLTESTW = number of W inner-iterations between tests for convergence

BPO = initial value of PSI in interior

BWO = " " " W " "

BP1, BP2, BP3, BP4 = initial values of PSI on right, top, left and bottom boundary lines, resp.

BW1, BW2, BW3, BW4 = initial values of W on right, top, left and bottom boundary lines, resp.

```
      PROGRAM NS4
      COMMON/1/M,N,MP,NP,XMIN,XMAX,YMIN,YMAX,H
      COMMON/INIT/BP0,BP1,BP2,BP3,BP4,BW0,BW1,BW2,BW3,BW4,M1,N1
      DIMENSION PSI(41,41),PSISAV(41,41),W(41,41),WSAV(41,41),AUX(41,41)
      TYPE INTEGER TOLTEST,TOLTESTP,TOLTESTW
C
C   READ IN NUMBER OF PROBLEMS.
      READ 903, NPROBS
 903  FORMAT(I5)
C
      DO 70 IPROB=1,NPROBS
C
C   READ IN PROBLEM PARAMETERS.
      READ 905, XMIN,XMAX,YMIN,YMAX,R,H,OMEGAP,OMEGAW,TOL,XI,DELTA
 905  FORMAT(4F5,F10,3F5,E10,2F5)
      READ 906, M,N,MP,NP,ITERMAX,TOLTEST,ITERMAXP,TOLTESTP,ITERMAXW,
     1 TOLTESTW
 906  FORMAT(10I5)
      READ 907, BP0,BP1,BP2,BP3,BP4,BW0,BW1,BW2,BW3,BW4
 907  FORMAT(10F5)
C
C   PRINT PROBLEM DESCRIPTION AND PARAMETERS.
      PRINT 909, XMIN,XMAX,YMIN,YMAX,N,M,H,OMEGAP,OMEGAW,TOL,TOLTEST,
     1 ITERMAX,R
 909  FORMAT(1H1,20X,13HPROBLEM NO. 4,10X,58HNAVIER-STOKES EQUATIONS FOR
     1 FLOW IN A CAVITY (UNIT SQUARE) //
     2 20X,12HRANGE OF X =,F6.2,3H TO,F6.2 /
     3 20X,12HRANGE OF Y =,F6.2,3H TO,F6.2 /
     4 20X,9HTHERE ARE,I5,15H HOROZONTAL AND,I5,56H VERTICAL LINES COMPR
     5ISING THE GRID, INCLUDING BOUNDARY./20X,15HSTEP SIZE (H) =,F10.5 /
     6 20X,39HRELAXATION FACTOR FOR STREAM FUNCTION =, F6.2 /
     6 20X,33HRELAXATION FACTOR FOR VORTICITY =,F6.2 /
     7 20X,11HTOLERANCE =,E10.1 / 19X,23H TOLERANCE TEST CYCLE =,I5/
     8 20X,30HMAXIMUM NUMBER OF ITERATIONS =,I6 /
     9 20X,17HREYNOLDS NUMBER =,F10.2 )
      PRINT 9091, ITERMAXP,TOLTESTP,ITERMAXW,TOLTESTW,XI,DELTA
 9091 FORMAT(20X,46HMAXIMUM ITERATIONS FOR PSI-ONLY CALCULATIONS =,I6 /
     1 20X,48HTOLERANCE TEST CYCLE FOR PSI-ONLY CALCULATIONS =I6 /
     2   20X,44HMAXIMUM ITERATIONS FOR W-ONLY CALCULATIONS =,I6 /
     3 20X,46HTOLERANCE TEST CYCLE FOR W-ONLY CALCULATIONS =I6 /
     4 20X,21HWEIGHT (XI) FOR PSI =,F6.2 /
     5 20X,22HWEIGHT (DELTA) FOR W =,F6.2 //)
C
C   COMPUTE ADDITIONAL PROBLEM PARAMETERS.
      M1=M-1
      N1=N-1
      M2=M-2
      N2=N-2
      RMAX=1.E+5
      H2=H*H
      CP1=1.0-OMEGAP
      CP2=0.25*OMEGAP
      CP5=.5*H
      CW00=1.0-OMEGAW
      CW5=2./H2
      CW7=2./H
```

```
      P2=0.5*R
      XI1=1.-XI
      DELTA1=1.0-DELTA
      ITER=ITOL=0
C
C   INITIALIZE VECTORS W AND PSI.
      CALL INIT(W,PSI)
C   PRINT INITIAL VECTORS W AND PSI.
      PRINT 911, ITER,RMAX
  911 FORMAT(///10X,16HAT ITERATION NO.,I6,20H  MAXIMUM RESIDUAL =,E12.4
     1 /20X,15HSTREAM FUNCTION)
      CALL PRTMAT(PSI)
      PRINT 912
 912  FORMAT(//20X,9HVORTICITY )
      CALL PRTMAT(W)
C
C   BEGIN MAIN LOOP FOR OUTER ITERATIONS.
C
C
 10   ITER=ITER+1
      ITOL=ITOL+1
      IF(ITOL .LT. TOLTEST) 105,102
C   SAVE STREAM FUNCTION FROM PREVIOUS OUTER ITERATION.
 102  DO 1021 J=2,N1
      DO 1021 I=2,M1
 1021 AUX(I,J)=PSI(I,J)
 105  RMAXP=1.E94
      ITERP=0
 106  ITOLP=0
 12   ITERP=ITERP+1
      ITOLP=ITOLP+1
C   COMPUTE ONE SWEEP OF INNER REGION FOR STREAM FUNCTION.
      DO 20 J=3,N2
      DO 20 I=3,M2
      PSISAV(I,J)=PSI(I,J)
 20   PSI(I,J)=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I.J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
      IF(ITOLP .LT. TOLTESTP) 24,25
C   RECALCULATE STREAM FUNCTION IN INNER REGION USING WEIGHTING.
 24   DO 241 J=3,N2
      DO 241 I=3,M2
 241  PSI(I,J)=XI*PSISAV(I,J)+XI1*PSI(I,J)
C   BOTTOM INNER BOUNDARY LINE FOR STREAM FUNCTION.
      DO 21 I=2,M1
 21   PSI(I,2)=.25*PSI(I,3)
C   LEFT AND RIGHT INNER BOUNDARY LINES FOR STREAM FUNCTION.
      DO 22 J=3,N2
      PSI(2,J)=.25*PSI(3,J)
 22   PSI(M1,J)=.25*PSI(M2,J)
C   TOP INNER BOUNDARY LINE FOR STREAM FUNCTION.
      DO 23 I=2,M1
 23   PSI(I,N1)=.25*PSI(I,N2)+CP5
      GO TO 12
C
 25   RMAX1P=0.0
C
```

```
C     RECALCULATE STREAM FUNCTION,USING WEIGHTING, AND RESIDUALS
C     ON INNER REGION.
      DO 207 J=3,N2
      DO 207 I=3,M2
      PSI(I,J)=XI*PSISAV(I,J)+XI1*PSI(I,J)
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
      IF(RES .GT. RMAX1P) 206,207
  206 RMAX1P=RES
  207 CONTINUE
C     BOTTOM INNER BOUNDARY LINE FOR STREAM FUNCTION AND RESIDUALS.
      DO 261 I=2,M1
      PSIOLD=PSI(I,2)
      PSINEW=.25*PSI(I,3)
      PSI(I,2)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 260,261
  260 RMAX1P=RES
  261 CONTINUE
C     LEFT AND RIGHT INNER BOUNDARY LINES FOR STREAM FUNCTION AND RESIDS.
      DO 281 J=3,N2
      PSIOLD=PSI(2,J)
      PSINEW=.25*PSI(3,J)
      PSI(2,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 270,271
  270 RMAX1P=RES
  271 PSIOLD=PSI(M1,J)
      PSINEW=.25*PSI(M2,J)
      PSI(M1,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 280,281
  280 RMAX1P=RES
  281 CONTINUE
C     TOP INNER BOUNDARY LINE FOR STREAM FUNCTION AND RESIDUALS.
      DO 291 I=2,M1
      PSIOLD=PSI(I,N1)
      PSINEW=.25*PSI(I,N2)+CP5
      PSI(I,N1)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 290,291
  290 RMAX1P=RES
  291 CONTINUE
      RMAXP=RMAX1P
C
C     TEST MAXIMUM RESIDUAL OF STREAM FUNCTION FOR DIVERGENCE.
      IF(RMAXP.GT. 1.E+5 ) 32,35
   32 PRINT 9017, RMAXP,ITERP
 9017 FORMAT(// 76H *****  DIVERGENCE IN PSI-ONLY ITERATIONS. PROBLEM AB
     1ANDONED. MAX RESIDUAL =E12.4,8H AT ITER,I6 )
      PRINT 9009
 9009 FORMAT(/ 20X,20HSTREAM FUNCTION, PSI )
      CALL PRTMAT(PSI)
      PRINT 9050
 9050 FORMAT(/ 20X,12HVORTICITY, W )
      CALL PRTMAT(W)
      GO TO 70
```

```
C     TEST MAXIMUM RESIDUAL OF STREAM FUNCTION FOR CONVERGENCE.
C     IF STREAM FUNCTION HAS CONVERGED FOR INNER ITERATIONS,
C     TEST FOR OUTER -ITERATION CONVERGENCE.
 35   IF(RMAXP.LE. TOL) 40,45
 40   PRINT 915, ITERP,TOL,RMAXP,ITER
 915  FORMAT(/    20H *****  AT ITERATION,I6,10H TOLERANCE,E10.1,
     1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6.10H  FOR PSI(,I5,1H))
      IF(ITOL.LT. TOLTEST) 50,405
 405  PRINT 9009
      CALL PRTMAT(PSI)
      IF(ITER .EQ.1) GO TO 425
      RMAX1=0.0
C     GET MAXIMUM OUTER -ITERATION RESIDUAL FOR STREAM FUNCTION.
      DO 42J=2,N1
      DO 42I=2,M1
      RES=ABSF(PSI(I,J)-AUX(I,J))
      IF(RES .GT. RMAX1) 41,42
 41   RMAX1=RES
 42   CONTINUE
      RMAX=RMAX1
 425  ITOL=0
      PRINT 9042, ITER,RMAX
 9042 FORMAT( / 20H *****  AT ITERATION,I6,20H  MAXIMUM RESIDUAL =,E15.6
     1 ,19H FOR LARGE PROBLEM. )
C     TEST OUTER-ITERATION RESIDUAL FOR DIVERGENCE.
      IF(RMAX .GT.1.E+5 ) 432,435
 432  PRINT 917, RMAX,ITER
 917  FORMAT(// 44H DIVERGENCE. RUN STOPPED WITH MAX RESIDUAL =,E12.4,
     1 8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C     TEST OUTER-ITERATION RESIDUAL FOR CONVERGENCE.
 435  IF(RMAX .LE. TOL) 440,445
 440  PRINT 9440, ITER,TOL,RMAX
 9440 FORMAT( /     20H *****  AT ITERATION,I6,10H TOLERANCE,E10.1,
     1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6 /)
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C     TEST IF MAXIMUM NUMBER OF OUTER ITERATIONS EXCEEDED.
 445  IF(ITER .GE. ITERMAX) 947,50
 947  PRINT 913, RMAX,ITER
 913  FORMAT(//57H *****  MAXIMUM NUMBER OF ITERATIONS USED. MAX RESIDUA
     1L =,E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
```

```
      GO TO 70
C   TEST IF MAXIMUM NUMBER OF INNER ITERATIONS EXCEEDED FOR STREAM FN.
 45   IF(ITERP.GE. ITERMAXP) 47,106
 47   PRINT 9013, RMAXP,ITERP
 9013 FORMAT(// 81H *****  MAXIMUM NUMBER OF ITERATIONS USED FOR PSI-ONL
     1Y ITERATIONS. MAX RESIDUAL =,E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C
C
C   BEGIN INNER-ITERATIONS FOR VORTICITY.
C
 50   RMAXW=1.E91
C   COMPUTE VORTICITY ON BOUNDARY USING WEIGHTING.
C   TOP AND BOTTOM BOUNDARIES.
      DO 5072 I=1,M
      W(I,N)=DELTA*W(I,N)-DELTA1*CW5*(PSI(I,N1)-H)
 5072 W(I,1)=DELTA*W(I,1)-DELTA1*CW5*PSI(I,2)
C   LEFT AND RIGHT BOUNDARIES.
      DO 5074 J=2,N1
      W(1,J)=DELTA*W(1,J)-DELTA1*CW5*PSI(2,J)
 5074 W(M,J)=DELTA*W(M,J)-DELTA1*CW5*PSI(M1,J)
      ITERW=0
 505  ITOLW=0
C   COMPUTE RELAXATION COEFFICIENTS FOR VORTICITY IN INTERIOR.
C   COMPUTE ONE SWEEP OF VORTICITY IN INTERIOR.
 506  ITERW=ITERW+1
      ITOLW=ITOLW+1
      DO 62 J=2,N1
      DO 62 I=2,M1
      WSAV(I,J)=W(I,J)
      A=PSI(I+1,J)-PSI(I-1,J)
      B=PSI(I,J+1)-PSI(I,J-1)
      R2A=R2*A
      R2B=R2*B
      IF(A .GE. 0.)51,55
 51   IF(B .GE. 0.)52,53
 52   CW0=4.0+R2A+R2B
      CW1=CW4=1.0
      CW2=1.+R2A
      CW3=1.0+R2B
      GO TO 60
 53   CW0=4.0+R2A-R2B
      CW3=CW4=1.0
      CW2=1.0+R2A
      CW1=1.0-R2B
      GO TO 60
 55   IF(B .GE. 0.)56,57
 56   CW0=4.0-R2A+R2B
      CW1=CW2=1.0
      CW3=1.0+R2B
      CW4=1.0-R2A
```

```
      GO TO 60
 57   CW0=4.0-R2A-R2B
      CW2=CW3=1.0
      CW1=1.0-R2B
      CW4=1.0-R2A
C
 60   CW0=OMEGAW/CW0
 62   W(I,J)=CW0C*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      IF(ITOLW .LT. TOLTESTW) 625,63
C   RECOMPUTE VORTICITY IN INTERIOR USING WEIGHTING.
 625  DO 628 J=2,N1
      DO 628 I=2,M1
 628  W(I,J)=DELTA*WSAV(I,J)+DELTA1*W(I,J)
      GO TO 506
C
C   RECOMPUTE VORTICITY,USING WEIGHTING, AND RESIDUALS IN INTERIOR.
 63   RMAX1W=0.0
      DO 65 J=2,N1
      DO 65 I=2,M1
      W(I,J)=WNEW=DELTA*WSAV(I,J)+DELTA1*W(I,J)
      RES=ABSF(WNEW-WSAV(I,J))
      IF(RES .GT.RMAX1W) 64,65
 64   RMAX1W=RES
 65   CONTINUE
      RMAXW=RMAX1W
C
C   TEST VORTICITY FOR INNER-ITERATION DIVERGENCE.
      IF(RMAXW.GT. 1.E+5 ) 665,666
 665  PRINT 9665, RMAXW,ITERW
 9665 FORMAT(// 55H *****  DIVERGENCE IN W-ONLY ITERATIONS. MAX RESIDUAL
     1 =,E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C   TEST VORTICITY FOR INNER-ITERATION CONVERGENCE
 666  IF(RMAXW .LE. TOL)67,675
 67   PRINT 9067, ITERW,TOL,RMAXW,ITER
 9067 FORMAT( /  20H *****  AT ITERATION,I6,10H TOLERANCE,E10.1,
     1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6,8H  FOR W(,I5,1H) /)
      IF(ITOL .EQ. 0) 673,10
 673  PRINT 9050
      CALL PRTMAT(W)
      GO TO 10
C   TEST IF MAXIMUM NUMBER OF INNER-ITERATIONS EXCEEDED FOR VORTICITY.
 675  IF(ITERW .GE.ITERMAXW) 677,505
 677  PRINT 9677 , RMAXW,ITERW
 9677 FORMAT(// 79H *****  MAXIMUM NUMBER OF ITERATIONS USED FOR W-ONLY
     1ITERATIONS. MAX RESIDUAL =,E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
```

```
      CALL PRTMAT(W)
C
C   END OF MAIN LOOP
 70   CONTINUE
      STOP
      END
```

REFERENCES

1. O. R. Burggraf, "Analytical and numerical studies of the structure of separated flows," J. Fluid Mech., 24, 1966, pp. 113 ff.

2. J. E. Fromm and F. H. Harlow, "Numerical solution of the problem of vortex street development," Phys. of Fluids, 6, 1963, pp. 975 ff.

3. D. Greenspan, "Numerical studies of two dimensional, steady state Navier-Stokes equations for arbitrary Reynolds number," Tech. Rpt. #9, Department of Computer Sciences, Univ. of Wisconsin, 1967.

4. C. E. Pearson, "A computational method for viscous flow problems," J. Fluid Mech.,21, 1965, pp. 611 ff.

5. L. M. Simuni, "Numerical solutions of several problems of motion of a viscous fluid," Eng. J. Acad. USSR, 4, 1964, pp. 446 ff (in Russian).

6. H. Schlichting, Boundary Layer Theory, McGraw-Hill, N. Y., 1960.

7. J. O. Hirschfelder, C. F. Curtiss and R. B. Bird, Molecular Theory of Gases and Liquids, Wiley, N. Y., 1954.

8. D. Greenspan, Introductory Numerical Analysis of Elliptic Boundary Value Problems, Harper and Row, N. Y., 1965.

9. W. E. Milne, Numerical Calculus, Princeton Univ. Press, Princeton, N. J., 1949.

10. G. K. Batchelor, "On steady laminar flow with closed streamlines at large Reynolds number," J. Fluid Mech., 1, 1956, pp. 177 ff.

11. J. Smith, "The coupled equation approach to the numerical solution of the biharminic equation by finite differences," Report supported by NASA Sustaining University Program - NGR 43-001-021, Univ. of Tennessee, n.d.